

# Dessin d'histogrammes

Vous voulez réaliser de très beaux histogrammes en trois dimensions tout en tapant un très petit programme ? Alors essayez donc "Dessins d'histogrammes"... Il s'agit d'un programme de quatre-vingts lignes en Basic, réalisé sur Amstrad 664. Outre les histogrammes, l'intérêt de ce programme réside dans les bandes FOR NEXT imbriquées et les utilisations de mode graphique et texte dans le même écran.

*Nicholas Seiersen*

```
1 CLS
2 LOCATE 10,10:PRINT "DESSIN D'HISTOGRAMMES "
3 LOCATE 13,12:PRINT "EN 3 DIMENSIONS"
4 LOCATE 10,15:PRINT "PAR NICHOLAS SEIERSEN"
5 FOR X=1 TO 2000:NEXT
10 MODE 1:INK 2,3:INK 3,6
20 f$(0)= "A"
30 f$(1)= "B"
40 f$(2)= "C"
50 f$(3)= "D"
60 f$(4)= "E"
70 f$(5)= "F"
80 f$(6)= "G"
90 f$(7)= "H"
105 INPUT "TITRE DU GRAPH";t$
110 INPUT "QUELLE EST LA PREMIERE ANNEE";y
```





```
120 INPUT "COMBIEN DE FAMILLES DE PRODUITS (MAX 6)";n:IF n>6 THEN 120
130 n=n-1
140 FOR i=0 TO 4 'rem boucles de saisie
150 FOR j=0 TO n
160 PRINT "YEAR ";y+i;" FAMILLE ";f$(j):INPUT r(i,j)
170 IF r(i,j)>m THEN m=r(i,j)
180 NEXT j
190 PRINT
200 NEXT i
210 MODE 2
220 FOR i=0 TO 4 'rem affichage des donnees
230 FOR j=0 TO n
240 PRINT i;r(i,j);
250 NEXT j::
260 PRINT:NEXT i
270 PRINT "le maximum est";m;" l'echelle est de 1:";m/16;
280 PRINT "voulez vous resaisir? o/n"
290 x$=INKEY$
300 IF x$="o" OR x$="O" THEN 160:ELSE IF x$="" THEN 290
310 MODE 1 'rem dessin de la grille
311 MOVE 128,144:DRAW 48,64:DRAW 48,304:DRAW 128,384:DRAW 608,384:DRAW 608,144:DRAW 528,64:DRAW 48,64
313 MOVE 128,384:DRAW 128,144
314 MOVE 48,176:DRAW 128,260:DRAW 608,260
315 MOVE 144,64:DRAW 224,144:DRAW 224,384
316 MOVE 240,64:DRAW 320,144:DRAW 320,384
317 MOVE 336,64:DRAW 416,144:DRAW 416,384
318 MOVE 432,64:DRAW 512,144:DRAW 512,384
319 MOVE 128,144:DRAW 608,144
330 FOR j=n TO 0 STEP -1 'rem dessin de l'arrete de l'histogramme
340 FOR i=4 TO 0 STEP -1
350 PEN 3:PAPER 0
360 y=21-INT((r(i,j)/m)*16)-i
370 x=4+i+6*j
380 LOCATE x,y
390 PRINT CHR$(214) 'dernier carre en perspective-triangle
400 FOR c=y+1 TO 21-i
410 LOCATE x,c:PRINT CHR$(143) 'carre plein
420 NEXT c
430 NEXT i
440 NEXT j
460 PEN 2 'rem dessin des cotes de l'histogramme
470 FOR j=0 TO n
480 FOR i=0 TO 4
490 PAPER 3
500 x=5+i+6*j
510 y=21-INT((r(i,j)/m)*16)-i
520 LOCATE x,y:PRINT CHR$(214) 'dernier carre en perspective-triangle
530 PAPER 0
540 LOCATE x,(21-i):PRINT CHR$(212) 'premier carre en
perspective-triangle inverse
550 FOR c=y+1 TO 20-i
560 LOCATE x,c:PRINT CHR$(143) 'carre plein
570 NEXT c
580 NEXT i
590 NEXT j
595 PEN 1
610 FOR j=0 TO n 'affichage des familles
620 LOCATE 4+6*j,22
630 PRINT f$(j)
640 NEXT
650 LOCATE 1,6 'affichage de l'echelle
660 PRINT INT(m)
670 LOCATE 1,14:PRINT INT(m/2)
680 LOCATE 1,21
690 PRINT " 0 "
700 LOCATE 4,23:PRINT t$
```